

PROGRAMA EN «BASIC» PARA EL CALCULO DE DL50 POR EL MÉTODO DE PROBITS

Ramiro Castro de la Mata¹ y Alfonso Zavaleta Martínez-Vargas^{2,3}

RESUMEN

Se describe un programa en lenguaje BASIC, para plataforma MSDOS, diseñado para calcular la Dosis Letal Media (DL_{50}) de agentes tóxicos utilizando el método de transformación en probits. Para la utilización del programa, el operador provee las dosis del agente tóxico empleado en el ensayo de toxicidad aguda letal, el número de animales utilizados en cada dosis, y el número de animales que murieron con dicha dosis; la DL_{50} , así como los límites fiduciales al 95% de confianza. Asimismo se presentan los valores definitivos y provisionales de la recta. El programa ofrece además la posibilidad de efectuar reanálisis de los datos en los casos en que se desee buscar mayor aproximación de los valores; gráfica, archivo de resultados en disco, y la opción de comparación de dos curvas, para la estimación de potencias relativas.

Palabras claves: DL50, Probits, DE50, Toxicidad aguda

ABSTRACT

A program in the language BASIC is described, for the MSDOS platform, designed to calculate the Average Lethal Dose (DL_{50}) of toxic agents used in the method of transformation in probits. For using this program, the operator supplies the dose of the toxic reagent used in the lethal acute toxicity test, the number of animals which die with this dose, DL_{50} , as well as the trustworthy limits to give 95% reliability. In addition final and provisional straight line values. The program offers in addition, the possibility of making a re-analysis of the data in the cases in which it is desired to seek a better approximation of the values, graphs, file the results on discs and the option to compare the two curves, for the estimation of relative strengths.

Key words: DL50, Probits, DE50, Acute toxicity

INTRODUCCION

Varios son los programas que se han empleado para la determinación de la Dosis Letal Media (DL_{50}) de agentes tóxicos. Entre ellos se cuentan métodos gráficos como el de Reed & Muench, y los métodos estadísticamente válidos como el método de Spearman Karber, y el método de transformación en probits, siendo ambos métodos aceptados por la Organización

Mundial de la Salud, y empleados rutinariamente en laboratorios dedicados a la evaluación de la toxicidad aguda en diferentes países^{2,3,5-7}. Dado que el empleo del método de transformación en probits, efectuado manualmente es laborioso y demanda el consumo de varias horas en los cálculos, se ha diseñado un programa de cómputo que pueda operar en computadoras pequeñas, y que reduzca significativamente el consumo de tiempo del operador.

El presente programa para calcular la DL_{50} por el método de transformación de porcentajes en probits se desarrolló inicialmente para una procesadora Timex-Sinclair 8080. Las Timex fueron las primeras computadoras de precio muy bajo, de US\$ 30, que aparecieron en el mercado a inicios de los ochentas, usando como monitor un receptor de Televisión, con una baja

¹ Médico Cirujano, Doctor en Medicina Humana, Profesor Principal, Sección Farmacología, Depto. Ciencias Fisiológicas, Universidad Peruana Cayetano Heredia, A.P. 4314, Lima 100, Perú.

² Médico Cirujano, Doctor en Farmacología, Profesor Principal, Sección Farmacología, Depto. Ciencias Fisiológicas, Universidad Peruana Cayetano Heredia, Lima, Perú.

³ Director General, Centro Nacional de Control de Calidad, Instituto Nacional de Salud, Lima, Perú.

velocidad de procesamiento y sólo 16 kilobytes de memoria. Ahora son miradas como un juguete primitivo pero en su tiempo, cuando aún no habían salido al mercado las computadoras personales, representaron la única posibilidad de efectuar operaciones muy complejas por las personas que no tenían acceso a las grandes máquinas de costo elevadísimo y que requerían locales especialmente acondicionados. El lenguaje usado por las Timex era el Basic y con él se desarrolló el primer programa que permitía reducir a menos de 7 minutos el cálculo de una DL_{50} . Este mismo programa ha sido mejorado, ampliado y perfeccionado para calcular además de los parámetros básicos, los niveles de ajuste de los resultados a las curvas teóricas, las gráficas y la impresión de los resultados.

El propósito de esta comunicación es el de presentar un programa para el cálculo de la DL_{50} que ha sido utilizado en la Sección Farmacología de la Universidad Peruana Cayetano Heredia, y en el Laboratorio Afiliado, del Instituto Nacional de salud para la estandarización de los venenos de serpientes y arácnidos peruanos, así como la evaluación de la potencia neutralizante de los sueros antiofídicos comerciales fabricados por el Instituto Nacional de Salud⁸⁻¹⁵.

MATERIALES Y METODOS

1) DISEÑO DEL PROGRAMA

Las fórmulas del cálculo de probits se han tomado de Burn y Col (1950)¹, Finney (1952)², las Tablas Científicas de la Ciba Geigy³, y Pearson y Hartley (1954)⁴. Cada una de ellas ha suministrado información que ha permitido la integración del programa, con el que se han recalculado los ejemplos y las tablas de las referencias para apreciar su precisión y ajuste

El programa tiene 356 instrucciones numeradas, pese a que la versión que usamos en el laboratorio corre en QBASIC y está compilada de tal modo de presentar los resultados en no más de uno o dos segundos después de ingresados los datos.

Debido a lo que se indicó anteriormente, los pasos 10 a 280 están dedicados a la prueba de normalidad por el método de ji cuadrado, que se prefirió añadir al comienzo para controlar mejor su ejecución y allí quedó.

Al inicio, en los pasos 290-520 se crea un arreglo (array) de 12 posiciones para mantener los datos ordenados y se efectúa un diálogo con el operador, pidiendo de manera ordenada los valores experimentales: Número de grupos, y en cada grupo dosis empleada, número de casos (número de animales del grupo) y número de positivos (número de animales muertos)

Conforme van ingresando los datos se colocan en el arreglo, directamente, transformados en logaritmos y coeficientes y posteriormente con todos sus valores de probits

De allí se pasa a los pasos 2080-2150 para ordenar los coeficientes de mortalidad en orden creciente. A los valores menores a 0.5 se les ha restado el valor de 1 para calcular el área bajo la curva normal para todos los valores por encima de 0.5. Si el coeficiente era menor, después se recupera su valor, pero ya sabiendo que área bajo la curva representa apartándose de la media. Para ello, cada grupo en el arreglo tiene un lugar que permite seguir su valor inicial. Esta estrategia se mantiene a lo largo de todo el programa. Se efectúa la iteración y cuando corresponde adjudicar el resultado a un valor determinado se le coloca en el espacio correspondiente de su arreglo. Luego se regresa a la posición inicial

En los pasos siguientes, 540 a 800 se calcula para cada grupo el valor del probit empírico, como área bajo la curva normal correspondiente al porcentaje (como coeficiente $\%/100$) obtenido. Para ello se sigue un proceso iterativo con intervalos de 0,005 hasta llegar al valor más lejano a 0,5 (50%)

A continuación, en los pasos 810-1020 se calculan los probits provisionales por aproximación por mínimos cuadrados de la línea recta de regresión del logaritmo de la dosis vs probit empírico recuperándose para cada log-dosis el probit que le corresponde en la línea ajustada, para ello se reordenan los valores y se eliminan momentáneamente los que corresponden a mortalidades de 0 y de 100%, los que se recuperan de la recta ajustada (pasos hasta 1280)

Los valores obtenidos se ajustan mediante los coeficientes de apreciación que permiten dar pesos variables a los diferentes valores de acuer-

do al número de casos y a su cercanía a los valores centrales obteniéndose los probits de trabajo con los que se calcula la regresión definitiva, en los pasos 1290 a 1620

En cada lugar del arreglo se han ido colocando los diversos valores de probits: empíricos, provisionales y de trabajo

Con los parámetros de la regresión de los probits de trabajo se pasa a la subrutina 30-280 donde se efectúa la prueba de ji cuadrado para apreciar la normalidad de los datos, (al 95% de precisión). La prueba se efectúa alternativamente para grados de libertad menores o mayores de 29. Si la curva no es normal, no se calculan los límites fiduciales

Los resultados se expresan como la ecuación de la recta definitiva y como serie de valores definitivos y provisionales. Se presentan juntos estos valores con el objeto de correrlos de modo iterativo si la desviación es muy grande, es decir, considerar los probits definitivos como provisionales y recalcular todo de nuevo hasta obtener un buen ajuste. Los límites fiduciales se calculan en función del logaritmo de la dosis al nivel del 95% considerando la distribución normal o con la fórmula alternativa de acuerdo a los resultados de la prueba ji cuadrado. Cuando la distribución es normal, el error estándar se presenta también para el logaritmo de la dosis. Un error bastante común, que incluso se aprecia en publicaciones de revisión muy exigente, es el de considerar el error estándar de la determinación en función de la dosis y no de su logaritmo y esto provoca una distorsión muy grande al calcular los límites. Por ejemplo una DL_{50} de 100 con error estándar de 10 daría como límites al 95%: los valores 80 a 120 (redondeando) Si se consideran los logaritmos, como debe ser, los límites son 10 y 1000, no solamente mucho mayores sino también asimétricos con respecto al valor central.

Las alternativas que se presentan con los resultados son las siguientes:

- 1.- Buscar mayor aproximación, de acuerdo al nivel de aproximación entre los probits provisionales y los definitivos
- 2.- Gráfica: La gráfica se calcula en los pasos 3410 a 3110 y se presenta en forma de cur-

va sigmoideal con los puntos experimentales. En muchas publicaciones se ha preferido el gráfico de la recta de probits, pero esto no ofrece ninguna ventaja y se debe a la dificultad tradicional de dibujar la sigmoideal ajustada a los datos, cosa que ya no ocurre con el uso de la computadora.

- 3.- Regresar al inicio para otros cálculos
- 4.- Imprimir los resultados: pasos 2160 a 2400 donde se arreglan los resultados para impresión en papel
- 5.- Archivar los resultados: pasos 3120 a 3410
Los resultados se guardan en un archivo con el objeto de poder comparar posteriormente varias series de datos en pruebas de toxicidad comparada haciendo el análisis correspondiente
- 6.- Comparar dos curvas: pasos 3420 al final, en pantalla se dan las instrucciones al operador y se encadena al programa correspondiente mediante el cual se comparan las DL_{50} y se establece la potencia relativa de los compuestos ensayados. El programa correspondiente sigue las fórmulas de Finney y se ha hecho de manera similar al presente.
- 7.- Terminar: Concluye el programa

Para la impresión de la gráfica no se ha hecho ninguna previsión en el programa ya que basta utilizar el comando «Print Screen» con la impresora en modalidad de gráficos y el comando «Graphics» en la memoria. (Hay que tener en cuenta que este comando tiene modificadores de acuerdo al tipo de impresora)

2) APLICACIÓN DEL PROGRAMA EN LA ESTIMACION DE LA DL_{50} DE VENENO DE ESCORPION

Veneno del escorpión peruano *Hadruroides lunatus* (Koch 1967) fue obtenido mediante electroestimulación y desecado en campana de vacío sobre $CaCl_2$. Alícuotas de veneno se diluyeron en volúmenes apropiados de NaCl 0,85% e inyectados (0,5 mL) por vía intraperitoneal en ratones albinos *mus musculus* (cepa Balb/C, ambos sexos, 20 a 24 g, 6 ratones/grupo, ámbito 32,47 a 378,96 mg/kg). Un

grupo de 6 ratones recibió 0,5 mL de NaCl 0.85% y utilizado como control. La mortalidad producida por el veneno fue registrada hasta las 72 horas. La DL_{50} fue estimada a las 72 horas, por el método de transformación en probits empleando los valores de dosis (mg/kg) , el número de ratones/grupo (casos) y el número de animales que murieron (positivos). Los datos experimentales utilizados para el análisis fueron obtenidos de Zavaleta (1980)¹⁷:

Dosis (mg/Kg)	Casos por grupo	Positivos (muertos)
32.47	4	1
60.31	5	2
111.02	5	3
204.37	4	4
378.96	5	5

RESULTADOS

J) El Programa de Probits. A continuación se lista el algoritmo completo del programa:

```

10 'probits v11
20 CLS : GOTO 290
30 IF AN > 29 THEN ZJ = ((XD / AN) ^ (1 / 3) -
(1 - 2 / (9 * AN))) / SQR(2 / (9 * AN)): GOTO
200
40 IF AN = 1 THEN ZJ = SQR(XD): GOTO 200
50 X1 = AN / 2: X5 = X1 + 5
60 G5 = SQR(2 * 3.14159 / X5) * (X5 ^ X5)
70 E5 = EXP(-(X5 - (1 / (12 * X5))) + (1 / (360 *
X5 * X5 * X5))))
80 G5 = G5 * E5
90 GJ = G5 / ((X1) * (X1 + 1) * (X1 + 2) * (X1 +
3) * (X1 + 4))
100 F1 = XD ^ (X1 - 1): F2 = (2 ^ X1) * GJ *
EXP(XD / 2): F1 = F1 / F2
110 CJ = 2 * XD / AN * F1: S1 = 1
120 FOR KJ = 1 TO 30: S2 = S1: S4 = 1
130 FOR J = 1 TO KJ: S4 = S4 * XD / (AN + 2 * J):
NEXT
140 S1 = S1 + S4
150 IF ABS(S1 - S2) < .0005 THEN KJ = 30
160 NEXT KJ
170 CJ = 1 - CJ * S1
180 RETURN
190 STOP
200 B1 = .319382: B2 = -.356564: B3 = 1.78148:
B4 = -1.82126: B5 = 1.33027

```

```

210 RJ = .231642
220 FK = SQR(1 / (2 * 3.14159))
230 XX = ABS(ZJ)
240 FJ = FK * EXP(-(XX ^ 2 / 2))
250 TJ = 1 / (1 + RJ * XX)
260 CJ = FJ * (B1 * TJ + B2 * TJ ^ 2 + B3 * TJ ^ 3
+ B4 * TJ ^ 4 + B5 * TJ ^ 5)
270 IF AN = 1 THEN CJ = 2 * CJ
280 RETURN
290 LOCATE 8
300 PRINT «ANALISIS DE PROBITS»: PRINT :
PRINT
310 INPUT « Nombre del compuesto (usar hasta 8
letras) «: MOC$
320 INPUT « Numero de grupos «: A
330 DIM A(A, 12)
340 FOR I = 1 TO A
350 PRINT «Dosis»: I, «: «:
360 INPUT A(I, 1)
370 PRINT «Numero de casos»: I, «»:
380 INPUT A(I, 2)
390 PRINT «Positivos»: I, «: «:
400 INPUT A(I, 3)
410 PRINT
420 NEXT I
430 FOR I = 1 TO A
440 A(I, 4) = A(I, 3) / A(I, 2)
450 A(I, 5) = LOG(A(I, 1)) / LOG(10)
460 A(I, 7) = A(I, 4)
470 IF A(I, 4) < .5 THEN A(I, 7) = 1 - A(I, 4)
480 A(I, 8) = I
490 NEXT I
500 CLS
510 LOCATE 8: PRINT «Espere unos instantes»
520 GOSUB 2080
530 S = A
540 M = 1 / (2 * 3.1415926536#) ^ .5
550 N = 2.71822818#
560 FOR I = 1 TO A
570 IF A(I, 7) = 1 THEN S = S - 1
580 NEXT I
590 J = I
600 E = .5
610 I = -.01
620 I = I + .01
630 IF I = 5.67 THEN GOTO 810
640 C = I + .01
650 R = M * N ^ (-I ^ 2 / 2)
660 P = M * N ^ (-(I + .005) ^ 2 / 2)
670 Q = M * N ^ (-C ^ 2 / 2)
680 D = .005 / 3 * (R + 4 * P + Q)
690 E = E + D
700 PRINT «*»:
710 IF A(J, 7) = .5 THEN A(A(J, 8), 6) = 5

```

```

720 IF A(J, 7) - E <= 0 THEN GOTO 740
730 GOTO 620
740 IF A(A(J, 8), 4) < .5 THEN A(A(J, 8), 6) = 5 -
    C
750 IF A(A(J, 8), 4) > .5 THEN A(A(J, 8), 6) = 5 +
    C
760 IF A = J THEN GOTO 810
770 J = J + 1
780 IF A(J, 7) = 1 THEN A(A(J, 8), 6) = 0
790 IF A(J, 7) = 1 THEN GOTO 760
800 GOTO 710
810 E = 0
820 F = 0
830 C = 0
840 D = 0
850 FOR I = 1 TO A
860 IF A(I, 6) = 0 THEN A(I, 5) = 0
870 F = F + A(I, 6)
880 E = E + A(I, 5)
890 C = C + A(I, 5) ^ 2
900 D = D + A(I, 6) * A(I, 5)
910 A(I, 5) = LOG(A(I, 1)) / LOG(10)
920 NEXT I
930 B = (D - E * F / S) / (C - E ^ 2 / S)
940 G = F / S - B * E / S
950 FOR I = 1 TO A
960 A(I, 9) = G + B * A(I, 5)
970 NEXT I
980 FOR I = 1 TO A
990 UU = A(I, 9) - 5
1000 A(I, 7) = ABS(UU)
1010 A(I, 8) = I
1020 NEXT I
1030 GOSUB 2080
1040 J = 1
1050 E = .5
1060 I = -.01
1070 I = I + .01
1080 IF I = 5.67 THEN GOTO 1290
1090 C = I + .01
1100 R = M * N ^ (-I ^ 2 / 2)
1110 P = M * N ^ (-(I + .005) ^ 2 / 2)
1120 Q = M * N ^ (-C ^ 2 / 2)
1130 D = .005 / 3 * (R + 4 * P + Q)
1140 E = E + D
1150 IF A(J, 7) = 0 THEN GOTO 1240
1160 IF A(J, 7) - C <= 0 THEN GOTO 1180
1170 GOTO 1070
1180 IF A(A(J, 8), 9) < 5 THEN A(A(J, 8), 10) = 1
    - E
1190 IF A(A(J, 8), 9) > 5 THEN A(A(J, 8), 10) = E
1200 A(A(J, 8), 11) = Q
1210 IF A = J THEN GOTO 1290
1220 J = J + 1
1230 GOTO 1150
1240 A(A(J, 8), 10) = .5
1250 A(A(J, 8), 11) = R
1260 IF J = A THEN GOTO 1290
1270 J = J + 1
1280 GOTO 1070
1290 FOR I = 1 TO A
1300 IF A(I, 4) = 0 THEN A(I, 12) = A(I, 9) - A(I,
    10) / A(I, 11)
1310 IF A(I, 4) = 1 THEN A(I, 12) = A(I, 9) + 1 -
    A(I, 10) / A(I, 11)
1320 A(I, 12) = A(I, 4) / A(I, 11) - A(I, 10) / A(I, 11)
    + A(I, 9)
1330 NEXT I
1340 J = 0
1350 L = 0
1360 K = 0
1370 O = 0
1380 Q = 0
1390 R = 0
1400 P = 0
1410 FOR I = 1 TO A
1420 J = J + A(I, 12)
1430 W = A(I, 2) * A(I, 11) ^ 2 / (A(I, 10) * (1 - A(I,
    10)))
1440 K = K + W
1450 L = L + W * A(I, 12)
1460 O = O + W * A(I, 12) ^ 2
1470 P = P + W * A(I, 5) * A(I, 12)
1480 Q = Q + W * A(I, 5)
1490 R = R + W * A(I, 5) ^ 2
1500 NEXT I
1510 SXX = R - Q ^ 2 / K
1520 SXY = P - Q * L / K
1530 XPRO = Q / K
1540 YPRO = L / K
1550 SY Y = O - L ^ 2 / K
1560 C = SXY / SXX
1570 H = L / K - C * Q / K
1580 DL = (5 - H) / C
1590 ES = 1 / C * (1 / K + ((DL - Q / K) * (DL - Q
    / K)) / (R - Q ^ 2 / K)) ^ .5
1600 XD = SY Y - C * SXY: AN = A - 2
1610 GOSUB 30
1620 PRINT «DATOS»: PRINT
1630 PRINT «Dosis», «Casos», «Positivos», «P.
    empíricos», «P. provisionarios»
1640 FOR I = 1 TO A
1650 PRINT A(I, 1), A(I, 2), A(I, 3), A(I, 6), A(I, 9)
1660 NEXT I
1670 PRINT: PRINT «RESULTADOS»
1680 PRINT «Provisional»; TAB(50); «Definitivo»
1690 PRINT «Logaritmo de DL50 «; TAB(30); (5 -
    G) / B; TAB(50); DL

```

```

1700 IF CJ <= .05 THEN PRINT «La distribución
no es normal»: GOTO 1720
1710 PRINT «Error standard «; TAB(50); ES
1720 PRINT «DL50»; TAB(30); EXP(((5 - G) / B)
* (LOG(10))); TAB(50); EXP(DL * (LOG(10)))
1730 PRINT «Regresión «; «Y=»; G; «+»; B; «X»;
TAB(50); «Y=»; H; «+»; C; «X»
1740 IF CJ > .05 THEN GOTO 1820
1750 BACC = C * SXY - 3.8416
1760 IF BACC <= 0 THEN GOTO 1840
1770 LIMSUP = XPRO + (1 / BACC) * (SXY * (5
- YPRO) + 1.96 * SQR(SXX * (BACC / K + (5
- YPRO) * (5 - YPRO))))
1780 LIMINF = XPRO + (1 / BACC) * (SXY * (5 -
YPRO) - 1.96 * SQR(SXX * (BACC / K + (5 -
YPRO) * (5 - YPRO))))
1790 PRINT «Limite Superior 95%»; TAB(50);
EXP(LIMSUP * LOG(10))
1800 PRINT «Limite Inferior 95%»; TAB(50);
EXP(LIMINF * LOG(10))
1810 GOTO 1840
1820 PRINT «Limite Superior 95%»; TAB(50);
EXP((DL + ES * 1.96) * LOG(10))
1830 PRINT «Limite Inferior 95%»; TAB(50);
EXP((DL - ES * 1.96) * LOG(10))
1840 PRINT
1850 PRINT «Para mayor aproximación, M «;
1860 PRINT «Para gráfica G»
1870 PRINT «Para volver a empezar V «;
1880 PRINT «Para imprimir I»
1890 PRINT «Para archivar los resultados A «;
1900 PRINT «Para comparar dos curvas C»
1910 PRINT « Para terminar T»
1920 SM$ = «»
1930 WHILE SM$ = «»: SM$ = INKEY$: WEND
1940 IF INSTR(«mM», SM$) THEN CLS : GOTO
2040
1950 IF INSTR(«tT», SM$) THEN CLS : CHAIN
«menu»
1960 IF INSTR(«cC», SM$) THEN CLS : GOTO
3420
1970 IF INSTR(«gG», SM$) THEN GOTO 2410
1980 IF INSTR(«aA», SM$) THEN GOTO 3120
1990 IF INSTR(«iI», SM$) THEN GOTO 2170
2000 IF INSTR(«Vv», SM$) THEN CLS : RUN
2010 GOTO 1930
2020 CLS
2030 GOTO 1600
2040 B = C
2050 G = H
2060 GOTO 950
2070 STOP
2080 FOR I = 1 TO (A - 1)
2090 IF A(I, 7) > A((I + 1), 7) THEN GOTO 2120
2100 NEXT I
2110 GOTO 2150
2120 SWAP A(I, 7), A(I + 1, 7)
2130 SWAP A(I, 8), A(I + 1, 8)
2140 GOTÓ 2080
2150 RETURN
2160 STOP
2170 INPUT «Titulo de la impresión»; A$
2180 LPRINT TAB(10); A$: LPRINT
2190 LPRINT «Dosis», «Casos», «Positivos», «Po-
sitivos %»
2200 FOR I = 1 TO A
2210 LPRINT A(I, 1), A(I, 2), A(I, 3), A(I, 4) * 100
2220 NEXT I
2230 LPRINT : LPRINT « RESULTADOS»
2240 LPRINT « Provisional»; TAB(50); «Definitivo»
2250 LPRINT «Logaritmo de DL50 «; TAB(30); (5
- G) / B; TAB(50); DL
2260 IF CJ <= .05 THEN PRINT «La distribución
no es normal»: GOTO 2280
2270 LPRINT «Error standard «; TAB(50); ES
2280 LPRINT «DL50»; TAB(30); EXP(((5 - G) / B)
* (LOG(10))); TAB(50); EXP(DL * (LOG(10)))
2290 LPRINT «Regresión «; «Y=»; G; «+»; B; «X»;
TAB(50); «Y=»; H; «+»; C; «X»
2300 IF CJ > .05 THEN GOTO 2370
2340 LPRINT «Limite Superior 95%»; TAB(50);
EXP(LIMSUP * LOG(10))
2350 LPRINT «Limite Inferior 95%»; TAB(50);
EXP(LIMINF * LOG(10))
2360 GOTO 2390
2370 LPRINT «Limite Superior 95%»; TAB(50);
EXP((DL + ES * 1.96) * LOG(10))
2380 LPRINT «Limite Inferior 95%»; TAB(50);
EXP((DL - ES * 1.96) * LOG(10))
2390 LPRINT
2400 GOTO 1850
2410 CLS : LOCATE 8
2411 INPUT «Titulo a poner a la grafica (Parte su-
perior)»; TIT$
2412 INPUT «Leyenda de la grafica (Parte inferior)
«; LEY$
2413 TIT = LEN(TIT$): TIT = (80 - TIT) / 2
2414 LEY = LEN(LEY$): LEY = (80 - LEY) / 2
2420 SCREEN 2
2430 XMN = 2.5: XMX = 7.5: YMN = 0: YMX = 1
2440 CLS
2450 NXTIC = 10: NYTIC = 10
2460 CLS
2470 SCREEN 2
2480 DSX = ABS(XMX - XMN): DSY =
ABS(YMX - YMN)
2490 SX = .1: SY = .1

```

```

2500 AXMN = XMN - DSX * SX: AXMX = XMX
      + DSX * SX
2510 AYMN = YMN + DSY * SY: AYMN = YMN
      - DSY * SY
2520 WINDOW (AXMN, AYMN)-(AXMX,
      AYMN)
2530 LINE (XMN, YMN)-(XMN, YMN)
2535 LINE (XMN, YMX)-(XMN, YMX)
2540 LINE (XMN, YMN)-(XMN, YMX)
2545 LINE (XMX, YMN)-(XMX, YMX)
2550 DXTIC = DSX * .02: DYTIC = DSY * .025
2560 XTIC = DSX / NXTIC: YTIC = DSY / NYTIC
2570 FOR I = 1 TO NXTIC
2580 XP = XMN + XTIC * I
2590 LINE (XP, YMN)-(XP, YMN + DYTIC)
2600 ROW = 24
2610 NEXT I
2620 FOR I = 1 TO NYTIC
2630 YP = YMN + YTIC * I
2640 LINE (XMN, YP)-(XMN + DXTIC, YP)
2650 NEXT I
2660 FOR I = 1 TO A
2670 Z = H + C * A(I, 5)
2680 CIRCLE (Z, A(I, 4)), DSX / 120
2690 NEXT I
2700 GRAF75 = (7.5 - H) / C: GRAF25 = (2.5 - H)
      / C: DGRAF = GRAF75 - GRAF25: DGRAF =
      DGRAF / 10
2710 FOR I = 1 TO NXTIC
2720 GRAF = GRAF25 + I * DGRAF
2730 GRAF = EXP(GRAF * LOG(10))
2740 XP = XMN + XTIC * I
2750 XC = PMAP(XP, 0)
2760 COL = INT(80 * XC / 640) - 1
2770 LOCATE 24, COL
2780 IF GRAF < .001 THEN PRINT USING
      «.###»; GRAF: : GOTO 2840
2790 IF GRAF < .01 THEN PRINT USING «.##»;
      GRAF: : GOTO 2840
2800 IF GRAF < 1 THEN PRINT USING «.##»;
      GRAF: : GOTO 2840
2810 IF GRAF < 10 THEN PRINT USING «.#.»;
      GRAF: : GOTO 2840
2820 IF GRAF < 100 THEN PRINT USING «##.»;
      GRAF: : GOTO 2840
2830 PRINT USING «#####»; GRAF: : GOTO 2840
2840 NEXT I
2850 FOR I = 1 TO NYTIC
2860 YP = YMN + I * YTIC
2870 YR = PMAP(YP, 1)
2880 ROW = CINT(24 * YR / 199) + 1
2890 LOCATE ROW, I
2900 PRINT USING «###.##»; YP
2910 NEXT I
2911 LOCATE 1
2912 PRINT SPC(TIT); TIT$
2913 LOCATE 25
2914 PRINT SPC(LEY); LEY$
2920 E = .5
2930 I = .01
2940 I = I + .01
2950 IF I > 2.45 THEN GOTO 3070
2960 R = M * N ^ (-I ^ 2 / 2)
2970 P = M * N ^ (-I + .005) ^ 2 / 2)
2980 Q = M * N ^ (-I + .01) ^ 2 / 2)
2990 D = .005 / 3 * (R + 4 * P + Q)
3000 E = E + D
3010 Y = (1 - E)
3020 J = 5 + I
3030 K = 5 - I
3040 PSET (J, E)
3050 PSET (K, Y)
3060 GOTO 2940
3070 AS = «»: WHILE AS = «»: AS = INKEYS:
      WEND
3080 SCREEN 0
3090 CLS
3100 LOCATE 8
3110 GOTO 1850
3120 PRINT «Drive en el que se va a archivar (A,
      B o C)»
3130 SM$ = «»
3140 WHILE SM$ = «»: SM$ = INKEYS: WEND
3150 SM$ = SM$ + «:»
3160 MOC$ = SM$ + MOC$
3170 CLOSE #1: OPEN MOC$ FOR OUTPUT
      AS #1
3180 PRINT #1, A
3190 FOR I = 1 TO A
3200 PRINT #1, A(I, 1)
3210 PRINT #1, A(I, 2)
3220 PRINT #1, A(I, 3)
3230 PRINT #1, A(I, 4)
3240 PRINT #1, A(I, 5)
3250 PRINT #1, A(I, 6)
3260 PRINT #1, A(I, 7)
3270 PRINT #1, A(I, 8)
3280 PRINT #1, A(I, 9)
3290 PRINT #1, A(I, 10)
3300 PRINT #1, A(I, 11)
3310 PRINT #1, A(I, 12)
3320 NEXT I
3330 PRINT #1, C
3340 PRINT #1, SYY
3350 PRINT #1, SXX
3360 PRINT #1, SXY
3370 PRINT #1, XPRO
3380 PRINT #1, YPRO

```

```

3390 PRINT #1, H
3400 PRINT #1, K
3410 GOTO 1840
3420 LOCATE 8
3430 PRINT « Primero hay que archivar los resulta-
dos»: PRINT
3440 PRINT « Si ya se archivaron oprimir Y»

3450 PRINT « Para archivar, oprimir A»
3460 SM$ = «»
3470 WHILE SMS = «»: SM$ = INKEY$: WEND
3480 IF INSTR(«aA», SM$) THEN GOTO 3120
3490 IF INSTR(«yY», SM$) THEN CHAIN
«COMPara»
    
```

2) ESTIMACION DE LA DL₅₀ DE VENENO DE ESCORPION


DATOS

Dosis	Casos	Positivos	P. empíricos	P. provisorios
32.47	4	1	4.32	4.058452
60.31	5	2	4.74	4.832025
111.02	5	3	5.26	5.594392
204.37	4	4	0	6.356772
378.96	5	5	0	7.128242

RESULTADOS

	Resultado Provisional	Resultado Definitivo
Logaritmo de DL ₅₀	1.83878	1.836007
Error standard		.1131409
DL ₅₀	68.98907	68.54998
Regresión	Y= .2896745 + 2.87673 X	Y= .4635525 + 2.975779 X
Limite Superior 95%		114.2256
Limite Inferior 95%		41.13875

Asimismo, con el programa se obtuvo el siguiente gráfico:



DISCUSION

Uno de los primeros pasos en la evaluación farmacológica de productos naturales y nuevos productos de síntesis, es la estimación de la Dosis Letal Media (DL_{50})^{7,14,9}. Sin embargo, muchas veces los cálculos estadísticos resultan tediosos, más aun cuando los métodos disponibles demandan de un consumo grande de tiempo del operador en los cálculos, incrementando la posibilidad de generación de errores⁷. Surgió en ese contexto, la necesidad de contar con un programa informático que permitiera disminuir el tiempo de cálculo, y que además permitiera efectuar ajustes sucesivos, así como incluyera la evaluación estadística de los datos experimentales obtenidos. Una necesidad adicional surge cuando es necesario comparar potencias relativas de diferentes lotes de la misma substancia, o comparar la potencia tóxica de diferentes productos. Un requerimiento similar se presenta en las plantas de fabricación farmacéutica y de productos biológicos, donde es necesario mantener una potencia similar en los diferentes lotes fabricados a través del tiempo. En este caso resulta de singular importancia el

mantener el registro de los datos, a fin de poder comparar dichas potencias sucesivamente en el tiempo contra el valor de potencia del estándar previamente definido.

Sin embargo, pese al tiempo transcurrido, en la actualidad no es fácil conseguir un programa que calcule el ajuste de datos cualitativos y el antiguo ejercicio que se hizo hace tantos años sigue siendo una buena herramienta de trabajo para los cálculos en nuestros laboratorios. La estructura del programa refleja la evolución del primitivo, con una estrategia de no modificar lo que funciona sino añadir los nuevos avances permitidos por el avance de la tecnología de las computadoras personales. Incluso hay segmentos que no se utilizan pero que han quedado de las versiones primitivas. Lo damos a conocer en su integridad, y sin hacer ningún esfuerzo cosmético, no como un modelo a seguir o a copiar sino por que contiene algunas ideas que pueden servir para elaborar programas más perfeccionados.

REFERENCIAS BIBLIOGRÁFICAS

1. Burn, J.H., Finney, D.J., y Goodwin, L.G. (1950). *Biological Standardization* (2a ed.). Oxford University Press, London.
2. Finney, D.J. (1952). *Biological Method in Biological Assay*. Ch.Griffin & Co., London.
3. Ciba-Geigy (Eds). (1955). *Documenta Geigy. Tablas Científicas*. 5a Ed. Basilea, Ciba Geigy.
4. Pearson, G., y Hartley, P (Eds) (1954). *Biometrika Tables for Statisticians*, Vol 1, Cambridge University Press, London,
5. Gene, J.A. y Robles, A. (1987). Determinación de la dosis letal 50% por el método de Spearman Karber. *Rev. Méd. Hosp. Nal. Niños Costa Rica* 1(22): 35-40.
6. Robles, A. y Gene, J.A. (1990). Determinación de la dosis letal 50% por el método de Spearman Karber. *Toxicalc*. Instituto Clodomiro Picado, Universidad de Costa Rica. mimeografiado, 7 p.
7. Castro De La Mata, R. (1996). Cap. IV: Toxicidad: Diseños experimentales y aplicación de programas estadísticos computarizados. *En: Diseño e Interpretación de Pruebas Farmacológicas en Control de Calidad*. Vol I. Zavaleta A, Salas M, Cabezas C, Carrillo C, Chang J. (Editores). Ministerio de Salud, Instituto Nacional de Salud. Red de Laboratorios de Control de Calidad de Medicamentos del Sector Salud. Serie de Documentos No. 3. Lima, Art. Lautrec. p:13 - 38.
8. Olascoaga, E.; Zavaleta, A. and Marsh, N. A. (1988). Preliminary studies of the effects of a peruvian snake *Bothrops pictus* (jergón of the coast) upon fibrinogen. *TOXICON*, 26(5): 501 - 504.
9. Zavaleta A, Salas M, Villegas L, Castillo J. (1998). *Farmacología de Venenos y Antivenenos de Serpiente*. Zavaleta A, Salas M, Cabezas C, Chang J, Carrillo C. (Editores). Ministerio de Salud, Instituto Nacional de Salud. Centro Nacional de Control de Calidad. Serie de Documentos No. 6. Lima, FIMART, 125 págs.
10. Zavaleta Martínez - Vargas, Alfonso. (1989) *Caracterización farmacológica del*

- veneno de la serpiente *Lachesis muta muta* «shushupe», Tesis Doctoral (Farmacología), Universidad Peruana Cayetano Heredia, Lima. 94 págs.
11. Becerra, P.; Zavaleta, A. (1993) Destoxificación del veneno de *Lachesis muta muta* (shushupe) mediante irradiación neutrónica. BOLETIN DE QUIMICA (LIMA), 59 (2): 69 - 74
 12. Castañon, Y.; Zavaleta, A.; Salas, M.; Pancorbo, Z.; Rodriguez, M. (1993) Efectos letal, hemorrágico, coagulante, fosfolipásico y hemolítico de los venenos de las serpientes peruanas *Lachesis muta muta* y *Bothrops barnetti*. BOLETIN DE QUIMICA (LIMA), 59 (1): 33 - 47
 13. Villegas, L.; Aguirre, E. y Zavaleta, A. (1993) Efectos de la liofilización sobre cuatro actividades biológicas del veneno de *Bothrops atrox* (Serpentes: Viperidae). Rev Biol Trop, 41(3): 851 - 853.
 14. Zavaleta A. (1980). Introducción al estudio del veneno de escorpión de los pedregales *Hadruroides lunatus* Koch 1867: método de extracción y acciones farmacológicas. Tesis Bachiller en Biología, Programa Académico de Ciencias y Filosofía, Universidad Peruana Cayetano Heredia, Lima. 43 págs.
 15. Incio Ruiz, L.; Incio Ruiz, R.; Zavaleta Martínez - Vargas, A. y Gutierrez, J. M. (1993) Toxicidad y neutralización de venenos ofídicos peruanos de los géneros *Bothrops* y *Lachesis* (Serpentes: Viperidae). Rev Biol Trop, 41(3): 351 - 357.